



Programiranje kroz aplikacije

Rad sa stringovima

Rad sa vremenom i datumima

Osnovno o stringovima

- String kao tip podataka služi za smeštanje teksta.
- Postoje stringovi promenljive i fiksne dužine.
- Stringovi promenljive dužine mogu sadržati i do dve milijarde (tačnije, 2^{31}) karaktera; zauzimaju 10 bajtova plus memoriju koja je potrebna za smeštanje stringa. Deklarišu se na sledeći način:

`Dim S as String`

- Stringovi fiksne dužine mogu sadržati od 1 do 64 000 karaktera (tačnije, 2^{16}) i oni zauzimaju samo memoriju potrebnu za smeštanje stringa. Ne deklarišu se kao nizovi, već na sledeći način:

`Dim S as String * 50`

- Dodela vrednosti `String` promenljivoj se vrši sa:

`S = "Divan dan 12.9.2011"`

Nadovezivanje stringova. vbCrLf i vbTab


- Za nadovezivanje stringova se mogu koristiti karakteri `+` i `&`.

```
S = "Divan " + "dan "  
S = S & "12.9.2011"
```

- Preporučuje se korišćenje `&` umesto `+`, jer je `+` i operator sabiranja.
- Pri formiranju/nadovezivanju stringova mogu se koristiti specijalni karakteri `vbCrLf` (prelazak u novi red) i `vbTab` (tab karakter).

- Na primer:

```
S = "aaa" & vbTab & "a" & vbCrLf  
S = S & "bb" & vbTab & "b" & vbCrLf  
S = S & "c" & vbTab & "c"
```



- Broj spejsova koji odgovara jednom tabu se podešava opcijom `Tab Width` na tabu `Editor` prozora `Tools / Options` u VBE.

Funkcija Mid

- Karakterima stringa se ne pristupa kao članovima niza (C, MATLAB).
- Za pristupanje pojedinim karakterima i podstringovima stringa, koristi se funkcija `Mid`.

- **`Mid(S, start, N)`** vraća string koji se sastoji od `N` karaktera stringa `S` počev od karaktera čija je pozicija `start`. Argument `N` je opcion i ako se ne navede ide se do kraja stringa. Na primer:

`Mid("Program", 2, 4)` vraća "rogr"

`Mid("Program", 2)` vraća "rogram"

- Izmena podstringova datog stringa se takođe vrši pomoću `Mid`. Na primer:

`Mid(S, 1, 3) = "123"`

menja prva tri karaktera stringa `S` podstringom "123".

Poređenje stringova

- Za poređenje stringova može poslužiti funkcija **StrComp**.
- **StrComp(s1, s2, nacin)** poredi stringove **s1** i **s2** i vraća **0** ako su jednaki, **-1** ako je **s1 < s2**, i **1** ako je **s2 < s1**.
- Argument **nacin** predstavlja način poređenja stringova i može biti **binaran** (**nacin = 0**), kad se pravi razlika između velikih i malih slova, i **tekstualan** (**nacin = 1**), kad se ne pravi razlika. Ako se **nacin** ne navede, podrazumevano je binarno poređenje. Na primer:
StrComp("VBA","vba",0) vraća **-1**
StrComp("VBA","vba",1) vraća **0**
- Stringovi se mogu porediti i operatorom ispitivanja jednakosti **=**, koji vrši binarno poređenje (podrazumevani način poređenja).
- Podrazumevani način poređenja se definiše navođenjem **Option Compare Text** ili **Option Compare Binary** na vrhu modula.

Funkcije Len, Str i Val

- Dužina stringa se dobija pomoću funkcije **Len**, tj. **Len(S)** vraća dužinu (broj karaktera) stringa **S**.
- Funkcija **Str** konvertuje broj **X**, argument funkcije, u odgovarajući string. Na primer:
Str(-412.56) vraća string **"-412.56"**
- Funkcija **Val** vraća broj sadržan u stringu **S**, argumentu funkcije. Čitanje broja započinje s leva i završava se kad se naiđe na prvi nenumerički karakter. **Val** prepoznaje tačku kao decimalni separator, ali ne i zarez, ignoriše tabove i blanko znake. Na primer:
Val("301*44") vraća broj **301**
Val("301.1*44") vraća broj **301.1**
Val("301..1*44") vraća broj **301**

Funkcije Left i Right

- Funkcija **Left(S, N)** vraća string koji se sastoji od prvih **N** karaktera stringa **S**. Na primer:
`Left("Dobar dan",4)` vraća string "Doba"
- `Left(S, N)` vraća isto što i `Mid(S, 1, N)`.
- Funkcija **Right(S, N)** vraća string koji se sastoji od poslednjih **N** karaktera stringa **S**. Na primer:
`Right("Dobar dan",3)` vraća string "dan"
- `Right(S, N)` vraća isto što i `Mid(S, Len(S)-N+1)`.
- Za razliku of `Mid`, funkcije `Left` i `Right` se **ne mogu koristiti za upis**, tj. naredbe tipa
`Right("Dobar dan",1) = "#"`
nisu dozvoljene!

Funkcije Replace, InStr i InStrRev

- Funkcija **Replace(str, s1, s2)** vraća string **str** kod koga je svaki podstring **s1** zamenjen podstringom **s2**. Na primer:
`Replace("Banana", "an", "ib")` vraća string `"Bibiba"`
- Funkcija **Replace** ima dodatne opcione argumente koji definišu poziciju u stringu **str** od koje se vrši pretraga, broj zamena koje treba izvršiti i način poređenja stringova. Pogledati Help.
- Funkcija **InStr(s1, s2)** vraća poziciju prve pojave stringa **s2** u stringu **s1**. Ukoliko ga nema, funkcija vraća **0**. Na primer:
`InStr("Banana", "ana")` vraća broj `2`
- Funkcija **InStrRev(s1, s2)** vraća poziciju poslednje pojave stringa **s2** u stringu **s1**. Ukoliko ga nema, funkcija vraća **0**. Na primer:
`InStrRev("Banana", "ana")` vraća broj `4`

Funkcije LCase i UCase

- Funkcija **LCase(str)** vraća string kod koga su velika slova konvertovana u mala, a ostali karakteri su neizmenjeni. Na primer:
LCase("#A1G2") vraća string **"#a1g2"**
- Funkcija **UCase(str)** vraća string kod koga su mala slova konvertovana u velika, a ostali karakteri su neizmenjeni. Na primer:
UCase("?b1c2") vraća string **"?B1C2"**

Funkcije LTrim, RTrim i Trim

- Funkcija **LTrim(str)** eliminiše spejsove kojima počinje string **str**. Na primer:
`LTrim(" Makro")` vraća string "Makro"
- Funkcija **RTrim(str)** eliminiše spejsove kojima se završava string **str**. Na primer:
`RTrim("Makro ")` vraća string "Makro"
- Funkcija **Trim(str)** eliminiše spejsove kojima počinje i završava se string **str**. Na primer:
`Trim(" Makro ")` vraća string "Makro"

Funkcije Split i Join

- Funkcija **Split(str, kar)** deli string **str** na niz stringova, pri čemu se deljenje vrši na karakteru **kar**. **Funkcija vraća niz stringova.**
- Funkcija **Join(niz, kar)** vraća string dobijenih spajanjem svih stringova iz niza **niz**, pri čemu se stringovi razdvajaju karakterom **kar**.
- U obe funkcije, ako se **kar** izostavi, podrazumeva se da je spejs.

```
Sub SplitJoin(S As String, K As String)
  Dim Niz() As String, I As Integer, novi As String
  Niz = Split(S, K)
  Debug.Print "Nakon funkcije Split"
  For I = LBound(Niz) To UBound(Niz)
    Debug.Print Niz(I)
  Next
  Debug.Print "Nakon funkcije Join"
  novi = Join(Niz, "%")
  Debug.Print novi
End Sub
```

SplitJoin "Danas je fin dan", " "



```
Nakon funkcije Split
Danas
je
fin
dan
Nakon funkcije Join
Danas%je%fin%dan
```

Rad sa kodovima karaktera

- Funkcija **Asc(str)** vraća ANSI kôd prvog karaktera u stringu **str**. Na primer:
`Asc("A")` vraća broj **65**
- Funkcija **AscW(str)** vraća Unicode kôd prvog karaktera u stringu **str**. Na primer:
`AscW("Žarko")` vraća broj **381**
- Funkcija **Chr(kodkar)** vraća karakter koji odgovara ANSI kôdu **kodkar**. Na primer:
`Chr(65)` vraća karakter **"A"**
- Funkcija **ChrW(kodkar)** vraća karakter koji odgovara Unicode kôdu **kodkar**. Na primer:
`ChrW(382)` vraća karakter **"ž"**

Rad sa slovima š, đ, č, ć i ž

- Za rad sa našim slovima koristimo funkciju `ChrW`. Jedino što treba da znamo su Unicode kodovi naših slova (tabela ispod).

| Slovo | Š | Đ | Č | Ć | Ž | š | đ | č | ć | ž |
|-------|-----|-----|-----|-----|-----|-----|-----|-----|-----|-----|
| Kôd | 352 | 272 | 268 | 262 | 381 | 353 | 273 | 269 | 263 | 382 |

- Na primer, string "Šećer" bi se definisao na sledeći način:
`S = ChrW(352) + "e" + ChrW(263) + "er"`
- Na ovaj način ne možemo prikazati ova slova u korisničkim formama. Da bi to uradili, sistemski moramo podesiti opciju **Language for non-Unicode programs** na Serbian (Latin). Ova opcija se nalazi na tabu `Advanced` prozora `Regional and Language Options` kod `Windows XP`, na tabu `Administrative` prozora `Regional and Language Options` kod `Windows Vista`, na tabu `Administrative` prozora `Region and Language` kod `Windows 7`, odnosno prozora `Region` kod `Windows 8`.
- Sa ovom opcijom omogućavamo unos naših slova u VBE prozor.

Like operator

- **Like** operator omogućava moćno i elegantno poređenje stringova. Primenjuje se u sledećem obliku:

`string Like obrazac`

gde `obrazac` predstavlja string koji može sadržati tri džoker znaka:

- ? menja bilo koji karakter,
- * menja proizvoljan broj karaktera, i
- # menja bilo koju cifru.

- Nekoliko primera:

| | |
|---|--------------------------|
| <code>"Konik" Like "Kon?k"</code> | vraća <code>True</code> |
| <code>"Konik" Like "Ko?k"</code> | vraća <code>False</code> |
| <code>"Konik" Like "K*"</code> | vraća <code>True</code> |
| <code>"2017" Like "####"</code> | vraća <code>True</code> |
| <code>"Konik 2017" Like "K*##17"</code> | vraća <code>True</code> |

Like operator (nastavak)

- **Like** dozvoljava i pretragu samo određenih karaktera, koji se navode u srednjim zagradama []. Na primer, izraz "Konik" Like "Kon[aei]k" vraća **True**, jer se na četvrtoj poziciji traže samo karakteri **a**, **e**, **i**.
- Opseg karaktera se takođe navodi u [], koristeći karakter **-**. Na primer:
 - [a-z] menja bilo koje malo slovo,
 - [A-Z] menja bilo koje veliko slovo,
 - [A-D] menja velika slova **A**, **B**, **C** i **D**.
- Prilikom korišćenja opsega, opseg mora navesti u rastućem redosledu. Na primer, zapis [Z-A] bi doveo do greške u izvršavanju.
- Ako ispred opsega stavimo znak uzvika, onda proveravamo da li karakter ne pripada datom opsegu. Na primer, sa [!a-z] ispitujemo da li dati karakter nije malo slovo.

Like operator (nastavak)

- Nekoliko primera:

"H" Like "[A-Z]" vraća True

"H" Like "[!A-Z]" vraća False

"H5N1" Like "H#[!C-E]#" vraća True

- Ako na vrhu modula stoji `Option Compare Text`, poređenje sa `[A-Z]` i `[a-z]` daje isti rezultat.
- Provera da li tekući karakter pripada jednom od više opsega se vrši nadovezivanjem opsega u `[]`. Na primer, provera da li je karakter slovo, veliko ili malo, se može vršiti na sledeći način:
"H" Like "[a-zA-Z]" vraća True
- Iako na prvi pogled deluje logično da se opsezi razdvoje zarezima, to ne treba raditi jer bi se onda i zarez uzimao u obzir prilikom pretrage.

Like operator – Prikaz specijalnih karaktera

- Videli smo da se kao specijalni karakteri u stringu obrascu koriste `[,]`, `*`, `#` i `?`. Postavlja se pitanje - Kako koristiti ove karaktere kao ostale karaktere?
- Odgovor – Potrebno ih je smestiti unutar zagrada `[]`.
- Primjer:
"H#?1" Like "H#[#][?]"# vraća True
- U prethodnom primeru, `[#]` u "H#[#][?]"# se odnosi na karakter #, a # na sve cifre. Slično, `[?]` se odnosi na karakter ?.

Rad sa vremenom i datumima

- VBA poseduje tip podataka **Date** koji radi sa vremenom i datumima. VBA može da radi sa datumima iz opsega 1.1.100–31.12.9999. i vremenom od 0:00:00 do 23:59:59.

- Dodeljivanje vrednosti **Date** promenljivoj se vrši pomoću karaktera taraba (#) kojim počinje i završava se vrednost. Na primer, datum 11.10.2017. bi se u **Date** promenljivu **Datum** uneo na bilo koji od sledećih načina:

Datum = #October 11, 2017#

Datum = #11/10/2017#

- Vreme se takođe unosi pomoću karaktera #. Na primer, **Date** promenljivoj **Vreme** se dodela vrši na sledeći način:

Vreme = #18:15:22#

- U Code prozoru, VBA konvertuje unešeni podatak u format koji je sistemski podešen.

Funkcije Date, Time i Now

- Funkcija **Date** vraća tekući datum.
- Smislene su operacije sabiranja i oduzimanja sa datumima. Na primer, datum koji je za 123 dana ispred današnjeg se dobija sa **Date + 123**
- Funkcija **Time** vraća tekuće vreme.
- Funkcija **Now** vraća tekući datum i vreme.
- Pomoću **Date** i **Time** naredbi se mogu menjati sistemski datum i vreme (**nije dozvoljeno u Windows 8 i nadalje**) i to na sledeći način:
Date = #11 October 2017#
Time = #12:23:34#

Funkcija DateDiff

- Funkcija **DateDiff** vraća broj proteklih vremenskih jedinica između dva datuma. Sintaksa je:

DateDiff(VremJed, Datum1, Datum2)

gde argument **VremJed** predstavlja vremensku jedinicu u kojoj merimo razliku ("yyyy" - godine, "m" - meseci, "w" - nedelje, "d" - dani, "h" - sati, "n" - minuti, "s" - sekunde), dok su **Datum1** i **Datum2** datumi između kojih se meri razlika.

- Na primer, broj proteklih dana između 1. septembra 1939. i danas je **DateDiff("d", #1 September 1939#, Date)**

Funkcija Format

- Funkcija `Format` je vrlo pogodan alat pri formatiranju numeričkih vrednosti, vremena i datuma. Mi ćemo koristiti njen skraćeni oblik:
`Format(izraz, format)`
gde se `izraz` formatira u skladu sa formatom `format`.
- Funkcija vraća string koji predstavlja formatirani izraz.
- Pri formatiranju izraza, možemo koristiti *predefinisane* i *korisničke* formate. Podrazumevani format zavisi od sistemskih podešavanja (Control Panel/Regional Options).

Predefinisani numerički formati

| Format | Opis |
|----------------|---|
| General Number | Podrazumevani format. Primer: <code>Format(12345,"General Number")</code> vraća "12345". |
| Currency | Prikazuje se separator hiljada, oznaka valute i dva decimalna mesta. Oznaka valute zavisi od sistemskog podešavanja. Primer: <code>Format(12345,"Currency")</code> vraća "12.345,00 €". |
| Fixed | Prikazuje se najmanje jedna cifra levo od decimalnog zareza i najmanje dve cifre desno od decimalnog zareza. Primer: <code>Format(1222.345,"Fixed")</code> vraća "1222,35". |
| Standard | Prikazuje se separator hiljada, najmanje jedna cifra levo od decimalnog zareza i najmanje dve cifre desno od zareza. Primer: <code>Format(1222.345,"Standard")</code> vraća "1.222,35". |
| Percent | Prikazuje se broj pomnožen sa 100, dve cifre desno od decimalnog zareza i karakter procenat desno od zareza. Primer: <code>Format(0.12345,"Percent")</code> vraća "12,35%". |
| Scientific | Standardni scientific zapis. Cifra najveće težine je prikazana levo od decimalnog zareza, i prikazuje se 2 do 30 decimalnih mesta praćeno sa "E" i eksponentom. Primer: <code>Format(12300,"Scientific")</code> vraća "1,23E+04". |

Predefinisani formati za datum i vreme

| Format | Opis |
|--------------|---|
| General Date | Podrazumevani format. |
| Long Date | Prikazuje se datum u skladu sa sistemski definisanim long date formatom (na primer, 3. decembar 2017). |
| Medium Date | Prikazuje se datum u skladu sa sistemski definisanim medium date formatom (na primer, 03-dec-17). |
| Short Date | Prikazuje se datum u skladu sa sistemski definisanim short date formatom (na primer, 3.12.2017). |
| Long Time | Prikazuje se vreme u skladu sa sistemski definisanim long time formatom (na primer, 12:32:45). |
| Medium Time | Prikazuje se vreme u skladu sa sistemski definisanim medium time formatom (na primer, 12:32). |
| Short Time | Prikazuje se vreme u skladu sa sistemski definisanim short time formatom (na primer, 12:32). |

Korisnički numerički formati

- Specijalni simboli koji se koriste pri definisanju korisničkih numeričkih formata su dati u tabeli ispod.

| Simbol | Opis |
|-----------------------|--|
| # | Jedna cifra. Ništa se ne prikazuje ako nema unosa. |
| 0 | Jedna cifra. Prikazuje nulu ako nema unosa. |
| . | Pozicija decimalne tačke. |
| , | Pozicija separatora hiljada. Označava poziciju samo prve hiljade. |
| % | Množi broj sa 100 (samo za prikaz) i dodaje karakter %. |
| E+ e+ E- e- | Prikaz brojeva u scientific formatu. E- i e- smeštaju znak - u eksponent; E+ i e+ smeštaju znak + u eksponent. |
| \$ () - + <space> | Prikazuje date karaktere. |
| \ | Umetanje karaktera koji dolazi posle \. |
| "text" | Umetanje teksta pod znacima navoda. |

Korisnički numerički formati (primeri)

`Format(6353.4, "##,##0.00")` vraća "6,353.40"

`Format(334.9, "###0.00 kg")` vraća "334.90 kg"

`Format(4, "0.00%")` vraća "400.00%"

`Format(6353.4, "00.00000E+")` vraća "63.53400E+2"

`Format(6353.4, "#.#E+")` vraća "6.4E+3"

`Format(1234567, "00-00-000")` vraća "12-34-567"

Primer 1

- Napisati funkciju **Obrnut** koja za argument ima string **S** i vraća string koji ima obrnut redosled karaktera u odnosu na **S**.

```
Function Obrnut(S As String) As String
    Dim I As Integer
    Obrnut = ""
    For I = 1 To Len(S)
        Obrnut = Mid(S, I, 1) & Obrnut
    Next
End Function
```

Primer 2

- Napisati funkciju **BezSlova** koja za argument ima string **S** i koja vraća string koji se dobija kada se iz ulaznog stringa izbace slova.

```
Function BezSlova(S As String) As String
    Dim I As Integer
    BezSlova = ""
    For I = 1 To Len(S)
        If Mid(S, I, 1) Like "[!A-Za-z]" Then
            BezSlova = BezSlova & Mid(S, I, 1)
        End If
    Next I
End Function
```

Primer 3

- Napisati funkciju **ObrnilmePrez** koja za argument ima string koji predstavlja ime i prezime studenta, razdvojeni spejsom, i koja treba da obrne redosled imena i prezimena (prvo prezime, pa ime).

```
Function ObrnilmePrez(S As String) As String
    Dim K As Integer
    K = InStr(S, " ")
    ObrnilmePrez = Mid(S, K + 1, Len(S) - K) & _
        " " & Mid(S, 1, K - 1)
End Function
```

Primer 4

- Napisati funkciju `DecUBin` koja za argument ima prirodan broj `N` i koja vraća string koji predstavlja binarni zapis broja `N`.

```
Function DecUBin(N As Integer) As String
    Dim Cifra as Integer
    DecUBin = ""
    Do Until N = 0
        Cifra = N Mod 2
        DecUBin = Cifra & DecUBin
        N = N \ 2
    Loop
End Function
```